

Description

`irtgraph icc` plots item characteristic curves (ICCs) for binary items and category characteristic curves (CCCs) for categorical items for the currently fitted IRT model.

Quick start

2PL model for binary items b1 to b10

```
irt 2pl b1-b10
```

Plot ICCs for all items

```
irtgraph icc
```

Plot ICCs and item difficulties for items b1, b5, and b9

```
irtgraph icc b1 b5 b9, blocation
```

GRM for ordinal items o1 to o5, items coded 1, 2, 3

```
irt grm o1-o5
```

Plot CCCs for selected item categories

```
irtgraph icc 1.o1 3.o5 2.o1
```

Plot CCCs for the first category of all items

```
irtgraph icc 1.o*
```

Fit a group 2PL model

```
irt 2pl b1-b9, group(female)
```

Plot ICCs for selected items and groups

```
irtgraph icc (b1) (0: b5) (1: b9)
```

Menu

Statistics > IRT (item response theory)

Syntax

Basic syntax

```
irtgraph icc [ varlist ] [ , options ]
```

Full syntax

```
irtgraph icc ([ #: ] varlist [ , plot_options ]) ([ #: ] varlist [ , plot_options ]) [ ... ]  
[ , options ]
```

varlist is a list of items from the currently fitted IRT model.
#: plots curves for the specified group; allowed only after a group IRT model.

options	Description
Plots	
<code>blocation[(line_options)]</code>	add vertical lines for estimated item difficulties
<code>plocation[(line_options)]</code>	add horizontal lines for midpoint probabilities
<code>bcc</code>	plot boundary characteristic curves for categorical items
<code>ccc</code>	plot category characteristic curves
<code>range(##)</code>	plot over $\theta = \#$ to $\#$
Line	
<code>line_options</code>	affect rendition of the plotted curves
Add plots	
<code>addplot(plot)</code>	add other plots to the ICC plot
Y axis, X axis, Titles, Legend, Overall	
<code>twoway_options</code>	any options other than by() documented in [G-3] <i>twoway_options</i>
Data	
<code>n(##)</code>	evaluate curves at # points; default is n(300)
<code>data(filename[, replace])</code>	save plot data to a file

plot_options	Description
<code>blocation[(line_options)]</code>	add vertical lines for estimated item difficulties
<code>plocation[(line_options)]</code>	add horizontal lines for midpoint probabilities
<code>bcc</code>	plot boundary characteristic curves for categorical items
<code>ccc</code>	plot category characteristic curves
<code>line_options</code>	affect rendition of the plotted curves

varlist may use factor-variable notation; see [U] 11.4.3 Factor variables.
line_options in *plot_options* override the same options specified in *options*.
collect is allowed; see [U] 11.1.10 Prefix commands.

Options

Plots

`blocation[(line_options)]` specifies that for each ICC, a vertical line be drawn from the estimated difficulty parameter on the x axis to the curve. The optional *line_options* specify how the vertical lines are rendered; see [G-3] *line_options*. This option implies option `bcc`.

`plocation[(line_options)]` specifies that for each ICC, a horizontal line be drawn from the midpoint probability on the y axis to the curve. The optional *line_options* specify how the horizontal lines are rendered; see [G-3] *line_options*. This option implies option `bcc`.

`bcc` specifies that boundary characteristic curves (BCCs) be plotted for categorical items. The ICCs for the individual item categories are plotted by default. This option has no effect on binary items.

`ccc` specifies that category characteristic curves (CCCs) be plotted for all items. This is the default behavior for categorical items. For binary items, this option will plot ICCs for both outcomes.

`range(# #)` specifies the range of values for θ . This option requires a pair of numbers identifying the minimum and maximum. The default is `range(-4 4)` unless the estimated difficulty parameters exceed these values, in which case the range is extended.

Line

line_options affect the rendition of the plotted ICCs; see [G-3] *line_options*.

Add plots

`addplot(plot)` allows adding more graph twoway plots to the graph; see [G-3] *addplot_option*.

Y axis, X axis, Titles, Legend, Overall

twoway_options are any of the options documented in [G-3] *twoway_options*, excluding `by()`. These include options for titling the graph (see [G-3] *title_options*) and for saving the graph to disk (see [G-3] *saving_option*).

Data

`n(#)` specifies the number of points at which the ICCs, CCCs, and BCCs are to be evaluated. The default is `n(300)`.

`data(filename [, replace])` saves the plot data to a Stata data file.

Remarks and examples

`irtgraph icc` plots ICCs for binary items and CCCs for categorical items after estimating the parameters of an IRT model using `irt`.

ICCs are also known as item response functions and item response curves.

CCCs are also known as category response functions, option response functions, operating characteristic curves, and category response curves.

For categorical items, `irtgraph icc` also plots BCCs, which are probability curves for crossing a boundary. BCCs are also known as “category boundary curves”.

`irtgraph icc` is very flexible, and the best way to learn its capabilities is through examples.

➤ Example 1: ICCs for binary outcomes

We continue with the model from [example 1](#) of [\[IRT\] irt 1pl](#). Recall that we fit a 1PL model to the nine binary items. Here we use `estat report` to rearrange the estimated IRT parameters sorted by item difficulty.

```
. use https://www.stata-press.com/data/r19/masc1
(Data from De Boeck & Wilson (2004))

. irt 1pl q1-q9
(output omitted)


. estat report, sort(b) byparm

One-parameter logistic model                                Number of obs = 800
Log likelihood = -4142.3516
```

	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
Discrim	.852123	.0458445	18.59	0.000	.7622695	.9419765
Diff						
q8	-2.413443	.1691832	-14.27	0.000	-2.745036	-2.08185
q3	-1.817693	.1399523	-12.99	0.000	-2.091994	-1.543391
q9	-1.193206	.1162054	-10.27	0.000	-1.420965	-.965448
q1	-.7071339	.1034574	-6.84	0.000	-.9099066	-.5043612
q2	-.1222008	.0963349	-1.27	0.205	-.3110138	.0666122
q4	.3209596	.0976599	3.29	0.001	.1295498	.5123695
q6	.6930617	.1031842	6.72	0.000	.4908243	.8952991
q7	1.325001	.1205805	10.99	0.000	1.088668	1.561335
q5	1.652719	.1329494	12.43	0.000	1.392144	1.913295


To plot ICCs for all items in the model, we simply type

```
. irtgraph icc
```




We can modify the default legend with the `legend()` option. Here, we shrink the legend and move it inside the plot region. We also specify a list of items explicitly so that the legend lists the ICCs in the order they appear in the graph. At the end of our interactive session, we came up with the following.

```
. irtgraph icc q8 q3 q9 q1 q2 q4 q6 q7 q5,
> legend(pos(4) ring(0) size(small) region(lcolor(black)))
```




Another way to present the ICCs is to turn off the legend and highlight the items with the lowest and highest estimated difficulty parameter. From the output of `estat report`, we see that item q8 has the lowest estimated difficulty parameter and item q5 has the highest estimated difficulty parameter. Here we put those items in a separate plotting group, request their estimated difficulty locations be plotted, and put the remaining items in another plotting group.

```
. irtgraph icc (q5 q8, blocation) (q1-q4 q6 q7 q9), legend(off)
```



This plot shows the information we wanted, but we can tinker further to make items q5 and q8 stand out and make the whole plot more dramatic. Here is the final result.

```
. irtgraph icc
> (q8 q5, lcolor(black) lwidth(thick) bloc(lcolor(black)))
> (q1-q4 q6 q7 q9, lpattern(dash)),
> range(-5 5) xlabel(-5 -2.41 0 1.65 5) legend(off) lcolor(red)
```



We admit the above works nicely for a 1PL model because the ICCs do not cross; 2PL and 3PL models may require a different approach, but the general idea remains the same—we rarely obtain the desired ICC plot on the first try and need to work incrementally to arrive at the graph that best suits the estimated model parameters.

◀

□ Technical note

For a binary item, it is standard practice to plot only the ICC for the probability of the positive outcome. Thus the following commands are equivalent.

```
. irtgraph icc q1
. irtgraph icc 1.q1
```

However, there are in fact two ICCs we could plot: one for the probability of the positive outcome and one for the probability of the negative outcome. To plot both ICCs, we can use any of the following:

```
. irtgraph icc 0.q1 1.q1
. irtgraph icc i.q1
. irtgraph icc q1, ccc
```

Because the two probabilities sum to 1, the ICC for the negative outcome is a mirror image of the ICC for the positive outcome reflected about the y axis at 0.5.

□

► Example 2: CCCs for categorical outcomes

We continue with the model introduced in [example 1](#) of [\[IRT\] irt grm](#). To easily present some graphical features, we collapse the last two categories into one for all items and refit the GRM.


```
. use https://www.stata-press.com/data/r19/charity
(Data from Zheng & Rabe-Hesketh (2007))

. recode ta1-ta5 (3=2)
(58 changes made to ta1)
(102 changes made to ta2)
(55 changes made to ta3)
(36 changes made to ta4)
(86 changes made to ta5)

. irt grm ta1-ta5
(output omitted)
```

For a model with many categorical items, we do not recommend using `irtgraph icc` without `varlist`, because the resulting graph will contain far too many plotted curves. With 5 items, each with 3 categories, the total number of CCCs in the default plot is 15. Here we focus on item `ta1`.


```
. irtgraph icc ta1, xlabel(-4 -1.45 1.20 4)
```



In a GRM, the adjacent probability curves do not cross at the estimated difficulty parameters. Each crossing point represents the level of the latent trait at which an examinee “transitions” from responding in one category versus the next. Thus, in the graph above, respondents whose trait level is below approximately -1.45 are most likely to answer 0 (strongly agree); respondents whose trait level is between approximately -1.45 and 1.20 are most likely to answer 1 (somewhat agree); and respondents whose trait level is above approximately 1.20 are most likely to answer 2 (somewhat or strongly disagree).

Because the GRM is defined in terms of cumulative probabilities, the estimated difficulties represent a point at which a person with $\theta = b_{ik}$ has a 50% chance of responding in category k or higher. We can use `irtgraph` to plot these probabilities with the corresponding estimated category difficulties. These probability curves are known as BCCs. We specify option `blocation`, which plots the category difficulties and also implies option `bcc`.

```
. irtgraph icc ta1, blocation
```



□ Technical note

In the example above, we typed

```
. irtgraph icc ta1
```

to plot the CCCs for item `ta1`. Because item `ta1` is coded 0, 1, or 2, we could have typed

```
. irtgraph icc 0.ta1 1.ta1 2.ta1
```

or

```
. irtgraph icc i.ta1
```

However, the first notation is most convenient to type. The factor notation comes in handy when we want to plot a particular category or change its appearance in the graph.

▷ Example 3: Combining graphs


Sometimes, it is useful to focus on one category and plot its CCCs for all items. Below we show one way of presenting this information. We do not label the curves, because we want to see only the overall shape and location of the CCCs. We could always play with the legend to identify the individual curves, as we did above in [example 1](#).


```

. irtgraph icc 0.ta*, legend(off) title("") lcolor(black) nodraw
> name(out0,replace)
. irtgraph icc 1.ta*, legend(off) title("") lcolor(red) nodraw
> name(out1,replace)
. irtgraph icc 2.ta*, legend(off) title("") lcolor(blue) nodraw
> name(out2,replace)
. graph combine out0 out1 out2, col(1) xsize(3) ysize(6)
> title("CCCs for items ta1-ta5")

```

CCCs for items ta1-ta5



► Example 4: ICCs for group IRT models

Here we demonstrate the behavior of `irtgraph icc` with group IRT models. The same comments apply to [\[IRT\] irtgraph iif](#). If you are not familiar with group IRT modeling, we suggest you read [\[IRT\] irt, group\(\)](#) first.


We use an abridged version of the mathematics and science data from [De Boeck and Wilson \(2004\)](#). We fit the following model.

```
. use https://www.stata-press.com/data/r19/masc2, clear
(Data from De Boeck & Wilson (2004))
. irt (0: 1p1 q1) (1: 1p1 q1) (1p1 q2 q3) (1: 1p1 q4 q5) (0: 1p1 q6),
> group(female)
(output omitted)
. estat greport
```

Parameter		Male	Female
q1	Discrim	1.430038	1.093494
	Diff	-.38922567	-.56959664
q2	Discrim	.82849958	.82849958
	Diff	.04043785	.04043785
q3	Discrim	.82849958	.82849958
	Diff	-1.743068	-1.743068
q6	Discrim	.99394585	
	Diff	.73061294	
q4	Discrim		.87968755
	Diff		.7142615
q5	Discrim		.87968755
	Diff		2.2134151
mean(Theta)		0	.02971286
var(Theta)		1	.98927041


Looking at the output of `estat greport`, we see that both groups were administered item q1 and that the parameters for this item differ between groups. In this situation, specifying `irtgraph icc q1` defaults to drawing the curves for both groups.

```
. irtgraph icc q1
```




If we wish to restrict `irtgraph icc` to drawing a curve for a specific group, we use a group identifier before the item. Here we draw the ICC for item q1 for the females only.

```
. irtgraph icc 1:q1
```




When item parameters are equal between groups, by default, `irtgraph icc` draws one ICC because the curve is the same for all groups. Here we graph the ICC for item q2.

```
. irtgraph icc q2
```




To convince yourself that the curves for item q2 are the same between groups, you can force `irtgraph icc` to draw separate curves for each group by putting each group in a separate equation.

```
. irtgraph icc (0:q2) (1:q2)
```



Now we can look at the ICC plot for all items.

```
. irtgraph icc
```



q1 is shown twice because the parameters are different between groups. q2 and q3 are each shown once because the parameters are group invariant. q4 and q5 were administered only to the female group, so they are each shown only for the female group. q6 was administered only to the male group, so it is shown only for the male group.



Stored results

`irtgraph icc` stores the following in `r()`:

Macros

<code>r(xvals)</code>	values used to label the x axis
<code>r(yvals)</code>	values used to label the y axis

References

- De Boeck, P., and M. Wilson, eds. 2004. *Explanatory Item Response Models: A Generalized Linear and Nonlinear Approach*. New York: Springer. <https://doi.org/10.1007/978-1-4757-3990-9>.
- Raciborski, R. 2015. Spotlight on irt. *The Stata Blog: Not Elsewhere Classified*. <https://blog.stata.com/2015/07/31/spotlight-on-irt/>.

Also see

- [IRT] [irt](#) — Introduction to IRT models
- [IRT] [irt 1pl](#) — One-parameter logistic model
- [IRT] [irt 2pl](#) — Two-parameter logistic model
- [IRT] [irt 3pl](#) — Three-parameter logistic model
- [IRT] [irt grm](#) — Graded response model
- [IRT] [irt hybrid](#) — Hybrid IRT models
- [IRT] [irt nrm](#) — Nominal response model
- [IRT] [irt pcm](#) — Partial credit model
- [IRT] [irt rsm](#) — Rating scale model
- [IRT] [irtgraph tcc](#) — Test characteristic curve plot

Stata, Stata Press, Mata, NetCourse, and NetCourseNow are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow is a trademark of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).